
1. Machine Description
The field of optics requires the grinding of precise pro-

files on the surface of a lens. An efficient method of per-
forming this task uses two different grinding bits simulta-
neously: coarse and fine (Figure 1). The first pass over the
lens material is a coarse cut, followed by a slightly delayed
precise cut. This could be done with two separate machines,
but a Galil controller has the ability to control two multi-
axis coordinated profiles simultaneously. In addition, the
lens must be spun on a platter at high RPM while the grind-
ing is occurring. The grinding bits themselves are high-
speed pneumatic spindles, controlled by digital output bits.

2. Requirements
This section summarizes the requirements for the

machine described above:

(1)Five servo axes (XYZWE)
(2)Index pulse, forward and reverse limit switches

for XYZW
(3)XY coordinated to follow any two-dimensional path 

specified by ASCII data files
(4)ZW coordinated to follow any two-dimensional path

specified by ASCII data files
(5)Two digital outputs for the two pneumatic grinding

bit spindles
(6)One digital input for the start button

Figure 1. Lens Grinder

3. Components Selected
This section describes the Galil hardware and software

products chosen to implement the machine’s control sys-
tem. Below is a complete bill of materials followed by a
description of major components.

Part number Description Unit Price (U.S)
Qty 1/Qty 100

DMC-2153-DC48 Ethernet Motion Controller 5-axis with 48V DC-DC converter $1395 / $915

AMP-20540 4-axis amplifier for driving four brush or brushless servos $795 / $495

BLM-N23-50-1000 or equivalent NEMA 23 Brushless Servo Motor, 1000 ppr encoder with Hall Sensors Consult manufacturer x4

CPS-12-56 or equivalent Power Supply 12A, 56V Consult manufacturer

ICM-20105 Direct connect breakout board with D-Sub connections for 5th axis $195 / $145

WSDK Servo Tuning Software Servo Tuning and Analysis Software $195 (one time)

Smart Terminal Controller communications software. Used for downloading arrays. Free

Table 1.  Bill of Materials for Lens Grinder Control System

Controller: DMC-2153-DC48

This machine was designed to have a local PC download
lens profiles to the controller. An Ethernet motion con-
troller is a good choice for this machine as it allows high
speed communications with the PC, while still being stand
alone to reside close to the motion axes. Therefore, the
DMC-2153 five axis Ethernet controller is selected. This
controller provides a 10BaseT Ethernet communications
port, 5 axes of motor control and on-board 48V DC-DC
converter.

Figure 2. DMC-2153-DC48

Amplifier: AMP-20540

To drive the four motors responsible for grinder posi-
tioning, we use the Galil AMP-20540 4-axis 500W brush-
less amplifier. This amplifier is connected directly to
controller axes XWZW via the
96-pin DIN connector.

Figure 3. AMP-20540 connects directly to 96-pin connector

Motor: BLM-N23-50-1000

For maintenance-free operation, we choose brushless
motors. Galil’s NEMA 23 #BLM-N23-50-1000 brushless
motors, or equivalent, are appropriate because all axes
require less than 0.3 Nm of continuous torque. Incremental
encoders with 1000 cycles per revolution are installed on
the motors resulting in 4000 quadrature counts per revolu-
tion. Hall sensors are not required on the motors as the
incremental encoders provide commutation tracks for in-
put to the amplifiers.

Interconnect:  ICM-20105

The ICM-20105 is an optically isolated signal break-
out board used to access the signals which drive the 5th
spindle axis (axis E). Because the velocity of the platter
needs to be held absolutely steady, the amplifier will
be a third-party velocity-mode unit. This ICM-20105
connects directly to the 96-pin DIN connector of the
DMC-2153-DC48.

Figure 4. Galil’s ICM-20105 connects directly to the
DMC-21x3 96-pin connector

4. Implementation
This section details how the components selected above

were used to implement the control system.

Vector Mode (CA, VM, VP)

The critical aspect of this machine is the control of two
independent coordinate systems. The application requires
that the XY axes be coordinated to follow a two-dimen-
sional path and that the ZW axes be simultaneously coordi-
nated to follow another two-dimensional path. XY are re-
ferred to as the S coordinate system and ZW are referred to
as T. Vector mode allows the two axes of each coordinate
system to be linked together to perform linear and circular
interpolation such that complex two-dimensional patterns
can be executed. Vector speeds anywhere along the path
can be specified to tailor the motion profile (see VP com-
mand < and > operators in command reference). The sim-
ple example below draws a square of 1000 counts per side
in vector mode:

CAS; VMXY ;'specify xy axes as S for vector mode

CAS; VP1000,0 ;'specify points to travel through

CAS; VP1000,1000

CAS; VP0,1000

CAS; VP0,0

CAS; VE ;'end the vector sequence

BGS ;'begin motion

AMS ;'wait until motion is complete



#AUTO ;'runs on controller power-up
'Before pressing the start button, the arrays are downloaded to the 
'controller from SmartTerminal’s File/Download Array menu option and
'burned into the controller’s EEPROM with the BV (burn variables) command
DM CoarseX[21], CoarseY[21], FineX[1001], FineY[1001] ;'define arrays

'axes are homed to the reverse limit switch, index pulse, and offset
'find limit switches
JG-10000, -10000, -10000, -10000 ;'jog towards reverse limit switch
BGXYZW ;'begin motion towards limit
AMXYZW ;'wait until we hit the limit

'find index pulses
JG500, 500, 500, 500 ;'move slowly towards the index pulse
FIXYZW ;'find index
BGXYZW ;'begin motion towards index
AMXYZW ;'wait until we hit the index

When the programmer is working with two distinct
coordinate systems, special care must be taken to ensure
that commands are being executed for the correct coordi-
nate system. This is controlled with the CA (Coordinate
Axes) command. CAS sets the S coordinate system and
CAT sets the T coordinate system. CAS; or CAT; should
be issued on the same line preceding the following com-
mands: VM, VP, CR, and VE. This is especially important
when calling these commands from multiple threads, where
the semicolon indicates that both commands on the line
are executed before switching to the next thread.

Program Description

After homing once upon power up, the program waits
for the start button. Once this is pressed,  the lens spindle
is started. Once the spindle is brought up to speed, the
coarse bit (S) begins cutting into the lens material for
some time, with the fine bit (T) following some time after.
This delay is managed by reading the vector segment
counter _CSS. The primary thread (thread 0) controls the
coarse cut, while a second thread (thread 1) controls the
fine cut.

Actual position data comes from the controller’s on-
board array space. These profiles were developed by
external equations that account for the tool diameter,
cut depth, and the desired refractive qualities of the fin-
ished lens. The arrays come from four files: CoarseX.txt,
CoarseY.txt, FineX.txt, and FineZ.txt. They are down-
loaded to the controller (before pressing the start button)
from SmartTerminal’s File/Download Array menu option

and burned into the controller’s EEPROM with the BV
(burn variables) command.

At the end of the two coordinate motion profiles, the
axes move away from the lens, the spindles are stopped
(the cycle is complete), and the program waits for the
start button to be pressed again.

Program Output

Figure 5 below shows the XYZW axis positions for a
complete grinding cycle.

Figure 5. The coarse (X vs. Y in blue) and fine (Z vs. W in
green) profiles captured with WSDK.

(Continued next page)

Program listing

The complete program (Lens.dmc) used for this article is included here. The comments explain the code in detail:



'perform offset
SP10000,10000,10000,10000 ;'set speed back to normal
PA 887, 52, 1611, 1437 ;'move to zero
BGXYZW ;'begin motion
AMXYZW ;'wait for motion complete
DP*=0                               ;'define position as zero

#Loop ;'loops once per lens
AI-1                              ;'wait for start button

'set spindle parameters
ACE=100000                        ;'set spindle acceleration
DCE=100000 ;'set spindle deceleration
JGE=100000                        ;'set spindle speed
BGE ;'begin turning spindle
ASE ;'wait until commanded speed is reached
WT1000                            ;'wait until spindle is steady

'move to first point on lens
SB1                               ;'enable coarse grinding bit
PAX=1000                          ;'set destination
BGX ;'begin motion
AMX ;'wait for motion complete

'set coarse grinder vector parameters
CAS; VMXY ;'set S plane coordinated axes to XY
VSS=2000                          ;'set S vector speed
VAS=100000                        ;'set S vector acceleration
VDS=100000                        ;'set S vector deceleration

XQ#Fine, 1                        ;'execute fine profile in thread 1

CAS; VP CoarseX[1], CoarseY[1]    ;'insert first point
BGS ;'begin vector motion

'loop that feeds array points to vector profiler
n = 2                             ;'initialize S plane counter
#CrsLoop

CAS; VP CoarseX[n], CoarseY[n]  ;'position data from array
n = n + 1                       ;'increment array counter
#CrsWait; JP#CrsWait, _LMS <= 0 ;'hold if buffer full

JP#CrsLoop, n <= 20               ;'perform 20 segments

CAS;VE ;'end S plane vector path
AMS ;'wait for motion complete

'return XY to Home
PAX=0                             ;'set destination
BGX ;'begin motion
AMX ;'wait for motion complete
PAY=0                             ;'set destination
BGY ;'begin motion
AMY ;'wait for motion complete
CB1                               ;'disable coarse grinding bit

#Wait; JP#Wait, _XQ1 = 1          ;'wait until thread 1 is done
STE ;'stop the spindle
AME ;'wait for spindle to stop

JP#Loop ;'go back to the beginning
EN ;'end thread 0

#Fine ;'thread that handle fine grinder
#Wait1; JP#Wait1, _CSS <= 2         ;'wait until coarse bit has advanced

'move to first point on lens
SB2 ;'enable fine grinding bit
PAZ=1000 ;'set destination
BGZ ;'begin motion
AMZ ;'wait for motion complete

'set fine grinder vector parameters
CAT; VMZW ;'set T plane coordinated axes to ZW
VST=1000 ;'set T vector speed
VAT=100000 ;'set T vector acceleration
VDT=100000 ;'set T vector deceleration

CAT; VP FineX[1], FineY[1] ;'insert first point
BGT ;'begin vector motion

'loop that feeds array points to vector profiler
m = 2 ;'clear T plane counter
#FinLoop

CAT; VP FineX[m], FineY[m] ;'position data from array
m  = m + 1 ;'increment array counter
#FinWait; JP#FinWait, _LMT <= 0 ;'hold if buffer full

JP#FinLoop, m <= 1000 ;'perform 1000 segments

CAT; VE ;'end T plane vector path
AMT ;'wait for motion complete

'return ZW to Home
PAZ=0 ;'set destination
BGZ ;'begin motion
AMZ ;'wait for motion complete
PAW=0 ;'set destination
BGW ;'begin motion
AMW ;'wait for motion complete
CB2 ;'disable fine grinding bit

EN ;'end thread 1

